
1 Introduction

In the last lecture we introduced Linear LF (LLF) as a substructural framework and
we saw how to represents proofs in the linear semi-axiomatic sequent calculus. But
it turns out that even some very simple systems of linear inference are difficult to
represent, particularly those with multiple conclusions that we used at the begin-
ning of the course.

In order to handle these we carefully extend the logical framework with some
positive types, leading us to Concurrent LF (CLF) [Watkins et al., 2002, Cervesato
et al., 2002, Watkins et al., 2004] which is implemented in the Celf language [Schack-
Nielsen and Schürmann, 2008, Schack-Nielsen, 2011]1. It turns out that this will
allow us to capture some concurrency in the computations we represent. We then
encode the dynamics of futures as an example that is significantly more direct than
possible in LLF.

2 Coin Exchange Revisited

Recall the rules for a linear coin exchange from Lecture 1, where q is a quarter, d is
a dime, and n is a nickel.

q

d d n
fromQ

d d n

q
toQ

d

n n
fromD

n n

d
toD

Here is a simple proof that from a quarter and a nickel we can get three dimes.
q

d d

n n

d
toD

fromQ

1https://github.com/clf/celf

In linear logic, we can internalize these as (structural!) propositions

q ⊸ (d⊗ d⊗ n)
(d⊗ d⊗ n) ⊸ q
d ⊸ (n⊗ n)
(n⊗ n) ⊸ d

We could make the second and the forth into something entirely negative by Cur-
rying and represent it in LLF, but not the other two.

fromQ : q ⊸ (d⊗ d⊗ n) ??
toQ : d ⊸ d ⊸ n ⊸ q
fromD : d ⊸ (n⊗ n) ??
toD : n ⊸ n ⊸ d

How can we solve this problem and represent this form of inference in LLF?
Think about it before you move on, because you actually have seen the technique
we need already (specifically in Lecture 2).

The technique is to move the inference steps into the antecedents, also flipping
their direction. The rules the are (with some arbitrary succedent C):

∆, d, d, n ⊢ C

∆, q ⊢ C
fromQ

∆, q ⊢ C

∆, d, d, n ⊢ C
toQ

∆, n, n ⊢ C

∆, d ⊢ C
fromD

∆, d ⊢ C

∆, n, n ⊢ C
toD

This we can represent in LLF, thinking of q, d, n, and C as judgments (and writing
c in lowercase), rather than propositions to avoid an extra level of indirection.

fromQ : (d ⊸ d ⊸ n ⊸ c)
⊸ (q ⊸ c)

toQ : (q ⊸ c)
⊸ (d ⊸ d ⊸ n ⊸ c)

fromD : (n ⊸ n ⊸ c)
⊸ (d ⊸ c)

toD : (d ⊸ c)
⊸ (n ⊸ n ⊸ c)

In functional programming this technique could be called continuation-passing style
where c stands for the continuation.

Now if we want to show that we can get three dimes from a quarter and a
nickel, it would be represented as

⊢ (d ⊸ d ⊸ d ⊸ c) ⊸ (q ⊸ c)

which we prove as follows:

...
d ⊸ d ⊸ d ⊸ c, d, d, d ⊢ c

d ⊸ d ⊸ d ⊸ c, d, d, n, n ⊢ c
toD

d ⊸ d ⊸ d ⊸ c, d, d, n, n ⊢ c
=

d ⊸ d ⊸ d ⊸ c, q, n ⊢ c
fromQ

⊢ (d ⊸ d ⊸ d ⊸ c) ⊸ (q ⊸ n ⊸ c)
⊸R× 3

The omitted part of the proof above is entirely straightforward.
If we represent this derivation as a term in LLF, it would be

⊢ λf. λq1. λn1. fromQ (λd1. λd2. λn2. toD (λd3. f d1 d2 d3) n1 n2) q1

where, as before, we write h (M1 ; . . .Mk) as h M1 . . . Mk.
There are few notes about this particular term representation. One is that the

coins have unique identities. For example, if we swap the arguments to f as in
f d3 d1 d2 we obtain a different term. In order to avoid this and reduce the number
of possible proofs, we can use proof irrelevance [Ley-Wild and Pfenning, 2007].

The other is that no matter what coins and exchange opportunities we have, the
proof term will always consist of nested constructors. One possible answer to this
is multifocusing [Chaudhuri et al., 2008]. Another is to explicitly construct a logical
framework with positive connectives, as we’ll do now.

3 CLF

The Concurrent Logical Framework was explicitly designed to allow natural en-
codings of linear forward inference and also the kind of concurrency from the pre-
vious example. We only give here a very brief description before we start using
it—you are referred to the technical reports and the implementation mentioned in
the introduction to the lecture for more information.

Below we have in red the LLF additions to LF and in blue the further additions
that CLF makes.

Negative types A,B ::= P | A→B | Πx : A.B(x) | A ⊸ B | A N B | {A+}
Positive types A+, B+ ::= 1 | A+ ⊗B+ | ∃x : A.B+(x) | A

Objects M ::= c S | x S | λx.M(x) | (M1,M2) | {E}
Spines S ::= M ; S | ( ) | π1 ; S | π2 ; S
Expressions E ::= . . .
Stable antecedents ∆ ::= · | ∆, xS : A | ∆, xL : A

There are a number of things to note here. Expressions E are the objects of positive
type, yet to be specified. The positive types are included in the negative ones as
{A+} which, nowadays, we would recognize as a form of shift. Similarly, the nega-
tive types are included directly in the positive ones (again, that should probably be
via a shift). Also, there are no positive atoms, even though there should be because
at the time we designed CLF we didn’t understand the type theory well enough.
Also, the antecedent A of A → B, A ⊸ B, and Πx : A.B(x) should be positive,
and probably also the A in ∃x : A.B(x). This last set of issues was recognized and
repaired in the design of Celf.

We omitted sums A ⊕ B because the branching nature of expressions compli-
cated the form of equality on terms we wanted to allow; a few further remarks later.
Before we come to the extension of objects/expressions, let’s write some types to
illustrate the use of positive types in the encoding of the coin exchange.

fromQ : q ⊸ { d⊗ d⊗ n }
toQ : d ⊸ d ⊸ n ⊸ { q }

fromD : d ⊸ {n⊗ n }
toD : n ⊸ n ⊸ { d }

here, we need to wrap even the singletons in the conclusions of the implications
because otherwise these clauses would be eligible for backchaining and not for
forward chaining.

Now to the proof terms for forward chaining. Since we do not have expressions
are just let bindings instead of general matches.

Expressions E ::= let {p} = M in E | T
Terms T ::= [ ] | [T1, T2] | [M,T ] | M
Patterns p ::= [ ] | [p1, p2] | [x, p] | x

We now show the CLF signature including the proof term for the previous example.

1 q : type.
2 d : type.
3 n : type.
4

5 toQ : d -o d -o n -o { q }.
6 fromQ : q -o { d * d * n }.
7 toD : n -o n -o { d }.
8 fromD : d -o { n * n }.
9

10 ex1 : q -o n -o { d * d * d } =
11 \q1. \n1. { let {[d1, [d2, n2]]} = fromQ q1 in
12 let { d3 } = toD n1 n2 in
13 [d1, d2, d3] }.

In this particular example, fromQ must come before toD because the argument n2

to toD is bound in the pattern that is matched against the result of fromQ.
More generally, though we consider two let expressions to be equivalent if they

can be swapped without any variable capture. That is,

(let p = M in let q = N in T )
= (let q = N in let p = M in T )

provided FV(p) ∩ FV(N) = ∅ = FV(q) ∩ FV(M)

This equality is baked into the definition of CLF at a fundamental level, just like the
renaming of bound variables. This allows is to think of the expressions as capturing
“true concurrency”, that is, different independent interleavings of concurrent actions
are indistinguishable.

It might be interesting to explore whether the notion of CBA-graphs we sketched
in the first two lectures would be an abstract representation of the quivalence
classes that arise from commuting independent actions. This may be related to
the notion of multifocusing mentioned earlier.

LECTURE NOTES DECEMBER 5, 2023



The Concurrent Logical Framework L22.6

4 Representing the Dynamics of Futures

Without further theory, which can be found in the given references, we show an
encoding of the dynamics of the positive fragment of linear SAX, which gives us
linear futures. This can easily be extended to encompass the whole language and
is much more abstract than the implementation in SML we used in this course.

1 val : type.
2 exp : type.
3 cont : type.
4

5 addr : type.
6

7 unit : val. % 1
8 pi1 : addr -o val. % A + B
9 pi2 : addr -o val.

10 pair : addr -o addr -o val. % A * B
11

12 unit_cont : exp -o cont.
13 plus_cont : (addr -o exp) & (addr -o exp) -o cont.
14 pair_cont : (addr -o addr -o exp) -o cont.
15

16 cut : (addr -o exp) -o (addr -o exp) -o exp.
17 id : addr -o addr -o exp.
18 write : addr -o val -o exp.
19 read : addr -o cont -o exp.
20

21 cell : addr -> val -> type.
22 proc : exp -> type.
23

24 pass : val -> cont -> exp -> type.
25 pass/unit : pass unit (unit_cont P) P.
26 pass/plus1 : pass (pi1 A) (plus_cont <(\x. P x), (\y. Q y)>) (P A).
27 pass/plus2 : pass (pi2 B) (plus_cont <(\x. P x), (\y. Q y)>) (Q B).
28 pass/pair : pass (pair A B) (pair_cont (\x. \y. P x y)) (P A B).
29

30 exec/cut : proc (cut (\x. P x) (\x. Q x))
31 -o { Exists a:addr. proc (P a) * proc (Q a) }.
32 exec/id : proc (id A B) -o cell B V -o { cell A V }.
33 exec/write : proc (write A V) -o { cell A V }.
34 exec/read : proc (read A K) -o cell A V -o pass V K P
35 -o { proc P }.

There are some subtle points here, such as the use of external choice in the encoding
of plus_cont, but in most respects it is an entirely straightforward representation.
Note also the use of the existential to create a fresh address dynamically when
executing a cut.

Unlike the coin exchange, we can actually execute SAX programs in this en-

coding because of the don’t care nondeterminism that underlies the dynamics. In
the coin exchange, we never reach quiescence, but here we do for terminating pro-
grams using futures.

In the first two examples we execute a SAX program for negation of a boolean
value represented by store with two cells.

1 #query * 1 * 1
2 Pi c0:addr. Pi c1:addr. Pi c2:addr.
3 cell c0 unit * cell c1 (pi1 c0)
4 * proc (read c1 (plus_cont (<(\u. write c2 (pi2 u)),
5 (\u. write c2 (pi1 u))>)))
6 -o { cell c0 unit * cell c2 (pi2 c0) }.
7

8 #query * 1 * 1
9 Pi c0:addr. Pi c1:addr. Pi c2:addr.

10 cell c0 unit * cell c1 (pi2 c0)
11 * proc (read c1 (plus_cont (<(\u. write c2 (pi2 u)),
12 (\u. write c2 (pi1 u))>)))
13 -o { cell c0 unit * cell c2 (pi1 c0) }.

The implementation will print a trace of the computations in the form of terms of
the given type. It shows structural bindings with \!x. M, while linear bindings are
just \x. M.

1 Query (*, 1, *, 1) ...
2 Solution: \!c0. \!c1. \!c2. \[X1, [X2, X3]]. {
3 let {X4} = exec/read X3 X2 pass/plus1 in
4 let {X5} = exec/write X4 in [X1, X5]}
5 Query ok.
6

7 Query (*, 1, *, 1) ...
8 Solution: \!c0. \!c1. \!c2. \[X1, [X2, X3]]. {
9 let {X4} = exec/read X3 X2 pass/plus2 in

10 let {X5} = exec/write X4 in [X1, X5]}
11 Query ok.

In the last query we use an existential quantifier in the succedent so as not to antic-
ipate the answer and let CLF’s forward inference compute it for us.

1 #query * 1 * 1
2 Pi c0:addr. Pi c1:addr. Pi c2:addr.
3 cell c0 unit * cell c1 (pi2 c0)
4 * proc (read c1 (plus_cont (<(\u. write c2 (pi2 u)),
5 (\u. write c2 (pi1 u))>)))
6 -o { Exists V0. Exists V2. cell c0 V0 * cell c2 V2 }.

Here, the answers !unit and !(pi1 c0) are presented in the results as witness for
the existentials. The exclamation mark shows that they are not linear.

1 Query (*, 1, *, 1) ..

2 Solution: \!c0. \!c1. \!c2. \[X1, [X2, X3]]. {
3 let {X4} = exec/read X3 X2 pass/plus2 in
4 let {X5} = exec/write X4 in [!unit, [!(pi1 c0), [X1, X5]]]}
5 Query ok.

